// File: lib/openzeppelin-contracts/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: lib/openzeppelin-contracts/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == _ENTERED;
    }
}

// File: lib/openzeppelin-contracts/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: lib/openzeppelin-contracts/contracts/token/ERC20/extensions/IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: lib/openzeppelin-contracts/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: lib/openzeppelin-contracts/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://consensys.net/diligence/blog/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: lib/openzeppelin-contracts/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: src/GearLiquidityBootstrapper.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.10;

import { Ownable } from "@openzeppelin/contracts/access/Ownable.sol";
import { IERC20 } from "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import { SafeERC20 } from "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import { ReentrancyGuard } from "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import { Address } from "@openzeppelin/contracts/utils/Address.sol";

import { IGearToken } from "./interfaces/IGearToken.sol";
import { ICurvePool } from "./interfaces/ICurvePool.sol";
import { ICurveFactory } from "./interfaces/ICurveFactory.sol";

import "./constants.sol";

enum Stage {
    INITIALIZED,
    GEAR_DEPOSIT,
    ETH_DEPOSIT,
    FAIR_TRADING,
    FINISHED,
    FAILED
}

contract GearLiquidityBootstrapper is Ownable, ReentrancyGuard {
    using SafeERC20 for IERC20;
    using Address for address;

    // EVENTS

    event StageChanged(Stage newStage);

    event GearCommitted(address indexed contributor, uint256 amount);

    event EthCommitted(address indexed contributor, uint256 amount);

    event PoolDeployed(address indexed pool);

    event PoolSeeded(uint256 initialGEAR, uint256 initialETH, uint256 initialLP);

    event GearSold(address indexed seller, uint256 amount, uint256 shearedAmount);

    event GearBought(address indexed buyer, uint256 amount);

    event GearTransfersUnlocked();

    event LPClaimed(address indexed contributor, uint256 amount);

    // GEAR DEPOSIT PARAMS

    // GEAR token
    IGearToken public constant gear = IGearToken(GEAR_TOKEN);

    // Max GEAR amount collected
    uint256 public constant gearMaxAmount = GEAR_MAX_AMOUNT;

    // Min GEAR amount collected
    uint256 public constant gearMinAmount = GEAR_MIN_AMOUNT;

    // GEAR amounts committed by contributors
    mapping(address => uint256) public gearCommitted;

    // Total amount of GEAR currently committed
    uint256 public totalGearCommitted;

    // Timestamp of GEAR deposit beginning
    uint256 public constant gearDepositStart = GEAR_DEPOSIT_START;

    // ETH DEPOSIT PARAMS

    // Max ETH amount collected
    uint256 public constant ethMaxAmount = ETH_MAX_AMOUNT;

    // Min ETH amount collected
    uint256 public constant ethMinAmount = ETH_MIN_AMOUNT;

    // ETH amounts committed by contributors
    mapping(address => uint256) public ethCommitted;

    // Total amount of ETH currently committed
    uint256 public totalEthCommitted;

    // Timestamp of ETH deposit beginning
    uint256 public constant ethDepositStart = GEAR_DEPOSIT_START + GEAR_DEPOSIT_DURATION;

    // FAIR TRADING PARAMETERS

    // Start of GEAR discounted selling
    uint256 public constant fairTradingStart = GEAR_DEPOSIT_START + GEAR_DEPOSIT_DURATION + ETH_DEPOSIT_DURATION;

    // End of GEAR discounted selling
    uint256 public constant fairTradingEnd = GEAR_DEPOSIT_START + GEAR_DEPOSIT_DURATION + ETH_DEPOSIT_DURATION + FAIR_TRADING_DURATION;
    
    // Duration of GEAR discounted selling
    uint256 public constant fairTradingDuration = FAIR_TRADING_DURATION;

    // Initial shearing percentage for GEAR selling (in 1e18 format)
    uint256 public constant shearingPctStart = STARTING_SHEARING_PCT;

    uint256 public constant SHEARING_PCT_DENOMINATOR = 10 ** 18;

    // FINISHED STAGE PARAMETERS

    // Total amount of LP tokens minted by the pool after liquidity deposit
    uint256 public totalLPTokens;

    // CURVE DEPLOY PARAMS

    // Address of the Curve factory for crypto pools
    ICurveFactory public constant curveFactory = ICurveFactory(CURVE_FACTORY);

    // WETH address (must be used as one of the Crypto pool assets)
    address public constant weth = WETH20;

    // Curve pool deployment parameters
    uint256 public constant curvePool_A = DEFAULT_A;
    uint256 public constant curvePool_gamma = DEFAULT_GAMMA;
    uint256 public constant curvePool_mid_fee = DEFAULT_MID_FEE;
    uint256 public constant curvePool_out_fee = DEFAULT_OUT_FEE;
    uint256 public constant curvePool_allowed_extra_profit = DEFAULT_ALLOWED_EXTRA_PROFIT;
    uint256 public constant curvePool_fee_gamma = DEFAULT_FEE_GAMMA;
    uint256 public constant curvePool_adjustment_step = DEFAULT_ADJUSTMENT_STEP;
    uint256 public constant curvePool_admin_fee = DEFAULT_ADMIN_FEE;
    uint256 public constant curvePool_ma_half_time = DEFAULT_MA_HALF_TIME;

    uint256 public constant PRICE_MULTIPLIER = 10 ** 18;

    // GLOBAL PARAMS

    // Current stage
    Stage public stage = Stage.INITIALIZED;

    // GEAR/ETH Curve pool (after it's deployed)
    ICurvePool public curvePool;

    constructor() {
        _transferOwnership(GEARBOX_TREASURY);
    }

    modifier withMinerSwitch(address newMiner) {
        address oldMiner = gear.miner();
        gear.setMiner(newMiner);
        _;
        gear.setMiner(oldMiner);
    }

    // ADMIN FUNCTIONS 

    function setGearMiner(
        address newMiner
    ) external onlyOwner {
        gear.setMiner(newMiner);
    }

    function fail() external onlyOwner {
        _fail();
    }

    function execute(address target, bytes memory data)
        external
        onlyOwner
        returns (bytes memory)
    {
        return target.functionCall(data);
    }

    // STAGE LOGIC

    modifier onlyStage(Stage requiredStage) {

        _advanceStage();

        require(
            stage == requiredStage,
            "Can't be called during the current stage"
        );
        _;
    }

    function _advanceStage() internal {

        Stage currentStage = stage;

        if (currentStage == Stage.FINISHED || currentStage == Stage.FAILED) {
            return;
        }

        if (currentStage == Stage.INITIALIZED) {
            if (block.timestamp >= gearDepositStart) {
                if (block.timestamp >= ethDepositStart) {
                    _fail();
                } else {
                    _startGearDeposit();
                }
            }
            return;
        } 
        
        if (currentStage == Stage.GEAR_DEPOSIT) {
            if (block.timestamp >= ethDepositStart) {
                if (block.timestamp >= fairTradingStart || totalGearCommitted < gearMinAmount) {
                    _fail();
                } else {
                    _startEthDeposit();
                }
            } 
            return;
        } 
        
        if (currentStage == Stage.ETH_DEPOSIT && block.timestamp >= fairTradingStart) {
            if (totalEthCommitted >= ethMinAmount) {
                if (block.timestamp >= fairTradingEnd) {
                    _finish();
                } else {
                    _startFairTrading();
                }
            } else {
                _fail();
            }
            return;
        } 
        
        if (currentStage == Stage.FAIR_TRADING && block.timestamp >= fairTradingEnd) {
            _finish();
        }
    }

    function advanceStage() external {
        _advanceStage();
    }

    function _startGearDeposit() internal {
        stage = Stage.GEAR_DEPOSIT;
        emit StageChanged(Stage.GEAR_DEPOSIT);
    }

    function _startEthDeposit() internal {
        stage = Stage.ETH_DEPOSIT;
        emit StageChanged(Stage.ETH_DEPOSIT);
    }

    function _startFairTrading() internal {
        _deployPool();
        _depositToPool();

        stage = Stage.FAIR_TRADING;

        emit StageChanged(Stage.FAIR_TRADING);
    }

    function _finish() internal {
        _unlockGEAR();

        stage = Stage.FINISHED;
        emit StageChanged(Stage.FINISHED);
    }

    function _fail() internal {
        stage = Stage.FAILED;
        emit StageChanged(Stage.FAILED);
    }

    function _deployPool() internal {
        uint256 initialPrice = totalGearCommitted * PRICE_MULTIPLIER / totalEthCommitted;

        address poolAddress = curveFactory.deploy_pool(
            "Curve GEAR/ETH",
            "crvGEARETH",
            [address(gear), address(weth)],
            curvePool_A,
            curvePool_gamma,
            curvePool_mid_fee,
            curvePool_out_fee,
            curvePool_allowed_extra_profit,
            curvePool_fee_gamma,
            curvePool_adjustment_step,
            curvePool_admin_fee,
            curvePool_ma_half_time,
            initialPrice
        );

        curvePool = ICurvePool(poolAddress);

        gear.approve(address(curvePool), MAX_INT);

        emit PoolDeployed(poolAddress);
    }

    function _depositToPool() internal withMinerSwitch(address(curvePool)) {

        uint256 totalEthCommitted_ = totalEthCommitted;
        uint256 totalGearCommitted_ = totalGearCommitted;

        totalLPTokens = curvePool.add_liquidity{value: totalEthCommitted_}([totalGearCommitted_, totalEthCommitted_], 0, true);

        emit PoolSeeded(totalGearCommitted_, totalEthCommitted_, totalLPTokens);
    }

    function _unlockGEAR() internal {
        gear.allowTransfers();
        gear.setMiner(address(gear));
        gear.transferOwnership(address(gear));

        emit GearTransfersUnlocked();
    }

    // GEAR DEPOSIT LOGIC

    function commitGEAR(uint256 amount) external onlyStage(Stage.GEAR_DEPOSIT) {

        uint256 totalGearCommitted_ = totalGearCommitted;

        if (totalGearCommitted_ + amount > gearMaxAmount) {
            amount = gearMaxAmount - totalGearCommitted_;
        }

        require(
            amount > 0,
            "Nothing to commit"
        );

        gear.transferFrom(msg.sender, address(this), amount);
        gearCommitted[msg.sender] += amount;
        totalGearCommitted += amount;

        emit GearCommitted(msg.sender, amount);
    }

    // ETH DEPOSIT LOGIC

    function commitETH() external payable nonReentrant onlyStage(Stage.ETH_DEPOSIT) {
        _commitETH();
    }

    receive() external payable nonReentrant onlyStage(Stage.ETH_DEPOSIT) {
        _commitETH();
    }

    function _commitETH() internal {

        uint256 amount = msg.value;
        uint256 totalEthCommitted_ = totalEthCommitted;

        if (totalEthCommitted_ + amount > ethMaxAmount) {
            amount = ethMaxAmount - totalEthCommitted_;
            payable(msg.sender).call{value: msg.value - amount}("");
        }

        require(
            amount > 0,
            "Nothing to commit"
        );

        ethCommitted[msg.sender] += amount;
        totalEthCommitted += amount;

        emit EthCommitted(msg.sender, amount);
    }

    // FAIR TRADING LOGIC

    function sellGEAR(uint256 amount, uint256 minETHBack) external onlyStage(Stage.FAIR_TRADING) withMinerSwitch(address(curvePool)) nonReentrant {

        gear.transferFrom(msg.sender, address(this), amount);

        uint256 currentShearingPct = getCurrentShearingPct();

        uint256 shearedAmount = amount * currentShearingPct / SHEARING_PCT_DENOMINATOR;
        uint256 amountToSell = amount - shearedAmount;

        curvePool.exchange(0, 1, amountToSell, minETHBack, true, msg.sender);

        emit GearSold(msg.sender, amountToSell, shearedAmount);
    }

    function buyGEAR(uint256 minGEARBack) external payable onlyStage(Stage.FAIR_TRADING) withMinerSwitch(address(curvePool)) nonReentrant {

        uint256 ethAmount = msg.value;

        uint256 gearAmount = curvePool.exchange{value: ethAmount}(1, 0, ethAmount, minGEARBack, true, msg.sender);

        emit GearBought(msg.sender, gearAmount);
    }

    function getCurrentShearingPct() public view returns (uint256) {
        if (block.timestamp < fairTradingStart) {
            return shearingPctStart;
        } else if (block.timestamp >= fairTradingEnd) {
            return 0;
        } else {
            return shearingPctStart * (fairTradingEnd - block.timestamp) / fairTradingDuration;
        }
    }

    function getETHFromGEARAmount(uint256 amount) external view returns (uint256) {

        if (stage < Stage.FAIR_TRADING) {
            return 0;
        }

        uint256 currentShearingPct = getCurrentShearingPct();
        uint256 amountToSell = amount - amount * currentShearingPct / SHEARING_PCT_DENOMINATOR;

        return curvePool.get_dy(0, 1, amountToSell);
    }

    function getGEARFromETHAmount(uint256 amount) external view returns (uint256) {
        return curvePool.get_dy(1, 0, amount);
    }

    // FINISHED LOGIC

    function claimLP() external onlyStage(Stage.FINISHED) {

        uint256 lpAmount = getPendingLPAmount();

        require(
            lpAmount != 0,
            "Nothing to claim"
        );

        gearCommitted[msg.sender] = 0;
        ethCommitted[msg.sender] = 0;

        IERC20(curvePool.token()).safeTransfer(msg.sender, lpAmount);

        emit LPClaimed(msg.sender, lpAmount);
    }

    function retrieveShearedGEAR() external onlyStage(Stage.FINISHED) onlyOwner {
        gear.transfer(owner(), gear.balanceOf(address(this)));
    }

    // FAILED LOGIC

    function retrieveGEAR() external onlyStage(Stage.FAILED) {

        uint256 amount = gearCommitted[msg.sender];
        gearCommitted[msg.sender] = 0;

        gear.transfer(msg.sender, amount);
    }

    function retrieveETH() external onlyStage(Stage.FAILED) nonReentrant {

        uint256 amount = ethCommitted[msg.sender];
        ethCommitted[msg.sender] = 0;

        payable(msg.sender).call{value: amount}("");
    }

    function takeGEARManagerBack() external onlyStage(Stage.FAILED) onlyOwner {
        gear.transferOwnership(owner());
    }

    // CONVENIENCE GETTERS

    function _getCurrentMinMaxAmounts() internal view returns(uint256 minGear, uint256 maxGear, uint256 minEth, uint256 maxEth) {

        uint256 totalGearCommitted_ = totalGearCommitted;
        uint256 totalEthCommitted_ = totalEthCommitted;

        minGear = totalGearCommitted_ <= gearMinAmount ? gearMinAmount : totalGearCommitted_;
        maxGear = block.timestamp >= ethDepositStart ? totalGearCommitted_ : gearMaxAmount;

        minEth = totalEthCommitted_ <= ethMinAmount ? ethMinAmount : totalEthCommitted_;
        maxEth = block.timestamp >= fairTradingStart ? totalEthCommitted_ : ethMaxAmount;
    }

    // Returns ETH / GEAR price range
    function getPriceRangeEthGear() external view returns (uint256 minPrice, uint256 maxPrice) {

        (uint256 minGear, uint256 maxGear, uint256 minEth, uint256 maxEth) = _getCurrentMinMaxAmounts();

        minPrice = minEth * PRICE_MULTIPLIER / maxGear;
        maxPrice = maxEth * PRICE_MULTIPLIER / minGear;
    }

    // Returns GEAR / ETH price range
    function getPriceRangeGearEth() external view returns (uint256 minPrice, uint256 maxPrice) {

        (uint256 minGear, uint256 maxGear, uint256 minEth, uint256 maxEth) = _getCurrentMinMaxAmounts();

        minPrice = minGear * PRICE_MULTIPLIER / maxEth;
        maxPrice = maxGear * PRICE_MULTIPLIER / minEth;
    }

    function getPendingLPAmount() public view returns (uint256 lpAmount) {

       uint256 totalLPTokens_ = totalLPTokens; 

       lpAmount = gearCommitted[msg.sender] * totalLPTokens_ / (2 * totalGearCommitted) + 
                           ethCommitted[msg.sender] * totalLPTokens_ / (2 * totalEthCommitted);
    }

    function getTimeUntilLPClaim() external view returns (uint256 lpAmount) {
        return fairTradingEnd > block.timestamp ? fairTradingEnd - block.timestamp : 0;
    }

}
// File: src/constants.sol



pragma solidity ^0.8.10;


address constant CURVE_FACTORY = 0xF18056Bbd320E96A48e3Fbf8bC061322531aac99;
address constant GEAR_TOKEN = 0xBa3335588D9403515223F109EdC4eB7269a9Ab5D;
address constant GEARBOX_TREASURY = 0x7b065Fcb0760dF0CEA8CFd144e08554F3CeA73D1;
address constant GEARBOX_TECH_MULTISIG = 0xA7D5DDc1b8557914F158076b228AA91eF613f1D5;
address constant WETH20 = 0xC02aaA39b223FE8D0A0e5C4F27eAD9083C756Cc2;

address constant DUMMY = 0xC4375B7De8af5a38a93548eb8453a498222C4fF2;
address constant DUMMY2 = 0xc4375B7dE8Af5a38A93548Eb8453A498222C4ff3;

uint256 constant MAX_INT = type(uint256).max;

uint256 constant GEAR_MAX_AMOUNT = 160_000_000 * 10 ** 18;
uint256 constant GEAR_MIN_AMOUNT = 120_000_000 * 10 ** 18;

uint256 constant ETH_MAX_AMOUNT = 3000 * 10 ** 18;
uint256 constant ETH_MIN_AMOUNT = 2000 * 10 ** 18;

uint256 constant GEAR_DEPOSIT_START = 1670918400;

uint256 constant GEAR_DEPOSIT_DURATION = 2 days;
uint256 constant ETH_DEPOSIT_DURATION = 2 days;

uint256 constant FAIR_TRADING_DURATION = 6 days;
uint256 constant STARTING_SHEARING_PCT = 25 * 10 ** 16;

uint256 constant DEFAULT_A = 400000;
uint256 constant DEFAULT_GAMMA = 145000000000000;
uint256 constant DEFAULT_MID_FEE = 26000000;
uint256 constant DEFAULT_OUT_FEE = 45000000;
uint256 constant DEFAULT_ALLOWED_EXTRA_PROFIT = 2000000000000;
uint256 constant DEFAULT_FEE_GAMMA = 230000000000000;
uint256 constant DEFAULT_ADJUSTMENT_STEP = 146000000000000;
uint256 constant DEFAULT_ADMIN_FEE = 5000000000;
uint256 constant DEFAULT_MA_HALF_TIME = 600;
// File: src/interfaces/ICurveFactory.sol



pragma solidity ^0.8.10;

interface ICurveFactory {
    function deploy_pool(
        string calldata _name,
        string calldata _symbol,
        address[2] calldata _coins,
        uint256 A,
        uint256 gamma,
        uint256 mid_fee,
        uint256 out_fee,
        uint256 allowed_extra_profit,
        uint256 fee_gamma,
        uint256 adjustment_step,
        uint256 admin_fee,
        uint256 ma_half_time,
        uint256 initial_price
    ) external returns (address);
}
// File: src/interfaces/ICurvePool.sol



pragma solidity ^0.8.10;

interface ICurvePool {
    function add_liquidity(
        uint256[2] memory amounts,
        uint256 min_mint_amount,
        bool use_eth
    ) external payable returns (uint256);

    function exchange(
        uint256 i,
        uint256 j,
        uint256 dx,
        uint256 min_dy,
        bool use_eth,
        address receiver
    ) external payable returns (uint256);

    function token() external view returns (address);

    function get_dy(uint256 i, uint256 j, uint256 dx) external view returns (uint256);

    function fee() external view returns (uint256);
}
// File: src/interfaces/IGearToken.sol



pragma solidity ^0.8.10;

import { IERC20 } from "@openzeppelin/contracts/token/ERC20/IERC20.sol";

interface IGearToken is IERC20 {
    function allowTransfers() external;

    function transferOwnership(address newManager) external;

    function setMiner(address _miner) external;

    function miner() external view returns (address);

    function manager() external view returns (address);

    function balances(int128 i) external view returns (uint256);

    function transfersAllowed() external view returns (bool);
}